

```

rmafterN:  move    $t0,$a0
             beqz    $t0,end

loop:        lw      $t1,4($t0)
             beq    $t1,$a1,look4next
             lw      $t0,($t0)
             bnez   $t0,loop
end:         jr      $ra

look4next:  lw      $t2,($t0)
             beqz   $t2,sew
nextloop:   lw      $t1,4($t2)
             bne   $t1,$a1,sew
             lw      $t2,($t2)
             bnez   $t2,nextloop
sew:        sw      $t2,($t0)
             move   $t0,$t2
             bnez   $t0,loop
             jr      $ra

```

ES2) In questo esercizio ci si doveva accorgere che il testo era (volutamente) mal posto. Non si specifica nulla riguardo alla possibile cancellazione della testa, tipo cancellare 2 da [2,5,3,2] e quindi, a rigor di logica, si deve presumere sia un caso possibile. In questa eventualità però l'indirizzo base della lista cambia e quindi nella specifica dell'esercizio va aggiunto di ritornare il (nuovo) indirizzo base in V0.

```

removeN:  move    $v0,$a0
             beqz   $v0,end
             move   $t0,$a0
             move   $t1,$zero

loop:        lw      $t2,4($t0)
             beq    $t2,$a1,look4next
             move   $t1,$t0
             lw      $t0,($t0)
             bnez   $t0,loop
end:         jr      $ra

look4next:  lw      $t0,($t0)
             beqz   $t0,sew
nextloop:   lw      $t2,4($t0)
             bne   $t2,$a1,sew
             lw      $t0,($t0)
             bnez   $t0,nextloop
sew:        beqz   $t1,headremove
             sw      $t0,($t1)
             bnez   $t0,loop
             jr      $ra

headremove: move   $v0,$t0
             bnez   $t0,loop
             jr      $ra

```

```

copy_list:  move    $t0,$a0          # sfrutto ampiamente L non vuota
              move    $v0,$a1
              move    $t1,$a1
loop:        lw     $t2,4($t0)
              sw     $t2,4($t1)
              lw     $t0,($t0)
              beqz   $t0,end
              addi   $t1,8
              sw     $t1,-8($t1)
              b      loop
end:        sw     $zero,($t1)
              jr     $ra

```

Un main fatto “come si deve”:

```

main:       addi    $sp,-4
              sw     $ra,0($sp)

              la     $a0,list
              la     $a1,freespace
              jal    copy_list
              move   $a0,$v0
              jal    print_list

              lw     $ra,0($sp)
              addi   $sp,4
              jr     $ra

```